# $Id: PKGBUILD 64752 2012-02-18 03:57:07Z svenstaro $
# Maintainer: Sven-Hendrik Haase <sh@lutzhaase.com>
# Contributor: jimmy-6
pkgname=libnoise
pkgver=1.0.0
pkgrel=2
pkgdesc="A portable, open-source, coherent noise-generating library for C++"
arch=('i686' 'x86_64')
url="http://libnoise.sourceforge.net/"
license=('LGPL')
makedepends=('cmake' 'unzip')
source=("http://prdownloads.sourceforge.net/libnoise/libnoisesrc-${pkgver}.zip"
        "CMakeLists.txt")
noextract=("libnoisesrc-${pkgver}.zip")
md5sums=('fc0d9b4f6477397568c3a9d5294b3b8c'
         '913a4145cdc7c38d43698386321ea85b')

build() {
  cd ${srcdir}
  unzip "libnoisesrc-${pkgver}.zip"
  cp 'CMakeLists.txt' 'noise'
  mkdir build
  cd build
  cmake ../noise \
    -DCMAKE_INSTALL_PREFIX=/usr \
    -DCMAKE_BUILD_TYPE=Release \
    -DCMAKE_SKIP_RPATH=ON
  make
}

package() {
  cd ${srcdir}/build
  make DESTDIR=${pkgdir} install
}
